
Generalized linear models

Generalize regular regression to non-normal data {(Yi , xi )}ni=1,
most often Bernoulli or Poisson Yi .

The general theory of GLMs has been developed to outcomes
in the exponential family (normal, gamma, Poisson, binomial,
negative binomial, ordinal/nominal multinomial).

The ith mean is µi = E (Yi )

The ith linear predictor is ηi = β0 + β1xi1 + · · ·+ βkxik = x′iβ.

A GLM relates the mean to the linear predictor through a link
function g(·):

g(µi ) = ηi = x′iβ.

14.1, 14.2 Binary response regression

Let Yi ∼ Bern(πi ). Yi might indicate the presence/absence of a
disease, whether someone has obtained their drivers license or not,
et cetera (pp. 555-556).

We wish to relate the probability of “success” πi to explanatory
covariates xi = (1, xi1, . . . , xik).

Yi ∼ Bern(πi ),

implying E (Yi ) = πi and var(Yi ) = πi (1− πi ).

Identity link g(µ) = µ

The identity link gives πi = β′xi . When xi = (1, xi )
′, this reduces

to
Yi ∼ Bern(β0 + β1xi ).

When xi large or small, πi can be less than zero or greater
than one.

Appropriate for a restricted range of xi values.

Can of course be extended to πi = β′xi where
xi = (1, xi1, . . . , xik).

Can be fit in SAS proc genmod.

Individual Bernoulli vs. aggregated binomial

Data can be stored in one of two ways:

If each subject has their own individual binary outcome Yi , we
can write model y=x1 x2 in proc genmod or proc
logistic.

If data are grouped, so that there are Y·j successes out of nj

with covariate xj , j = 1, . . . , c , then write model y/n=x1 x2.
This method is sometimes used to reduce a very large number
of individuals n to a small number of distinct covariates c .

5 / 59



Association between snoring and heart disease

From Agresti (2013).

Let s be someone’s snoring score, s ∈ {0, 2, 4, 5}.

Heart disease Proportion
Snoring s yes no yes
Never 0 24 1355 0.017
Occasionally 2 35 603 0.055
Nearly every night 4 21 192 0.099
Every night 5 30 224 0.118

This is fit in proc genmod:

data glm;

input snoring disease total @@;

datalines;

0 24 1379 2 35 638 4 21 213 5 30 254

;

proc genmod; model disease/total = snoring / dist=bin link=identity;

run;

proc genmod output

The GENMOD Procedure

Model Information

Description Value

Distribution BINOMIAL

Link Function IDENTITY

Dependent Variable DISEASE

Dependent Variable TOTAL

Observations Used 4

Number Of Events 110

Number Of Trials 2484

Criteria For Assessing Goodness Of Fit

Criterion DF Value Value/DF

Deviance 2 0.0692 0.0346

Pearson Chi-Square 2 0.0688 0.0344

Log Likelihood . -417.4960 .

Analysis Of Parameter Estimates

Parameter DF Estimate Std Err ChiSquare Pr>Chi

INTERCEPT 1 0.0172 0.0034 25.1805 0.0001

SNORING 1 0.0198 0.0028 49.9708 0.0001

SCALE 0 1.0000 0.0000 . .

Extracting useful inferences

The fitted model is

π̂(s) = 0.0172 + 0.0198s.

For every unit increase in snoring score s, the probability of heart
disease increases by about 2%.

The p-values test H0 : β0 = 0 and H0 : β1 = 0. The latter is more
interesting and we reject at the α = 0.001 level. The probability of
heart disease is strongly, linearly related to the snoring score.

We’ll denote the maximum likelihood estimates by β̂ instead of b
in this chapter. Both PROC LOGISTIC and PROC GENMOD give
MLEs.

14.2, 14.3 Logistic regression

Often a fixed change in x has less impact when π(x) is near zero
or one.

Example: Let π(x) be probability of getting an A in a statistics
class and x is the number of hours a week you work on homework.
When x = 0, increasing x by 1 will change your (very small)
probability of an A very little. When x = 4, adding an hour will
change your probability quite a bit. When x = 20, that additional
hour probably wont improve your chances of getting an A much.
You were at essentially π(x) ≈ 1 at x = 10.

Of course, this is a mean model. Individuals will vary.

logit link gives logistic regression

The most widely used nonlinear function to model probabilities is
the logit link:

logit(πi ) = log

(
πi

1− πi

)
= β0 + β1xi .

Solving for πi and then dropping the subscripts we get the
probability of success (Y = 1) as a function of x :

π(x) =
exp(β0 + β1x)

1 + exp(β0 + β1x)
.

When β1 > 0 the function increases from 0 to 1; when β1 < 0 it
decreases. When β = 0 the function is constant for all values of x
and Y is unrelated to x .

The logistic function is logit−1(x) = ex/(1 + ex).

Logistic functions for various β0 and β1

Logistic curves π(x) = eβ0+β1x/(1 + eβ0+β1x) with
(β0, β1) = (0, 1), (0, 0.4), (−2, 0.4), (−3,−1). What about
(β0, β1) = (log 2, 0)?

Logistic regression on snoring data

To fit the snoring data to the logistic regression model we use the
same SAS code as before (proc genmod) except specify
LINK=LOGIT and obtain b0 = −3.87 and b1 = 0.40 as maximum
likelihood estimates.

Criteria For Assessing Goodness Of Fit

Criterion DF Value Value/DF

Deviance 2 2.8089 1.4045

Pearson Chi-Square 2 2.8743 1.4372

Log Likelihood . -418.8658 .

Analysis Of Parameter Estimates

Parameter DF Estimate Std Err ChiSquare Pr>Chi

INTERCEPT 1 -3.8662 0.1662 541.0562 0.0001

SNORING 1 0.3973 0.0500 63.1236 0.0001

SCALE 0 1.0000 0.0000 . .

You can also use proc logistic to fit binary regression models.

proc logistic; model disease/total = snoring;

Logistic output

Response Variable (Events): DISEASE

Response Variable (Trials): TOTAL

Number of Observations: 4

Link Function: Logit

Response Profile

Ordered Binary

Value Outcome Count

1 EVENT 110

2 NO EVENT 2374

Model Fitting Information and Testing Global Null Hypothesis BETA=0

Intercept

Intercept and

Criterion Only Covariates Chi-Square for Covariates

AIC 902.827 841.732 .

-2 LOG L 900.827 837.732 63.096 with 1 DF (p=0.0001)

Score . . 72.688 with 1 DF (p=0.0001)

Analysis of Maximum Likelihood Estimates

Parameter Standard Wald Pr > Standardized Odds

Variable DF Estimate Error Chi-Square Chi-Square Estimate Ratio

INTERCPT 1 -3.8662 0.1662 541.0562 0.0001 . .

SNORING 1 0.3973 0.0500 63.1237 0.0001 0.384807 1.488

The fitted model is π̂(x) = exp(−3.87+0.40x)
1+exp(−3.87+0.40x) . As before, we reject

H0 : β1 = 0; there is a strong, positive association between snoring
score and developing heart disease. 13 / 59



Crab mating (Agresti, 2013)

Data on n = 173 female horseshoe crabs.

C = color (1,2,3,4=light medium, medium, dark medium,
dark).

S = spine condition (1,2,3=both good, one worn or broken,
both worn or broken).

W = carapace width (cm).

Wt = weight (kg).

Sa = number of satellites (additional male crabs besides her
nest-mate husband) nearby.

We are initially interested in the probability that a female
horseshoe crab has one or more satellites (Yi = 1) as a function of
carapace width.

Crab data in SAS

data crabs;

input color spine width satell weight @@; weight=weight/1000; color=color-1;

y=0; if satell>0 then y=1; id=_n_;

datalines;

3 3 28.3 8 3050 4 3 22.5 0 1550 2 1 26.0 9 2300 4 3 24.8 0 2100

4 3 26.0 4 2600 3 3 23.8 0 2100 2 1 26.5 0 2350 4 2 24.7 0 1900

3 1 23.7 0 1950 4 3 25.6 0 2150 4 3 24.3 0 2150 3 3 25.8 0 2650

3 3 28.2 11 3050 5 2 21.0 0 1850 3 1 26.0 14 2300 2 1 27.1 8 2950

3 3 25.2 1 2000 3 3 29.0 1 3000 5 3 24.7 0 2200 3 3 27.4 5 2700

3 2 23.2 4 1950 2 2 25.0 3 2300 3 1 22.5 1 1600 4 3 26.7 2 2600

5 3 25.8 3 2000 5 3 26.2 0 1300 3 3 28.7 3 3150 3 1 26.8 5 2700

5 3 27.5 0 2600 3 3 24.9 0 2100 2 1 29.3 4 3200 2 3 25.8 0 2600

3 2 25.7 0 2000 3 1 25.7 8 2000 3 1 26.7 5 2700 5 3 23.7 0 1850

3 3 26.8 0 2650 3 3 27.5 6 3150 5 3 23.4 0 1900 3 3 27.9 6 2800

4 3 27.5 3 3100 2 1 26.1 5 2800 2 1 27.7 6 2500 3 1 30.0 5 3300

4 1 28.5 9 3250 4 3 28.9 4 2800 3 3 28.2 6 2600 3 3 25.0 4 2100

3 3 28.5 3 3000 3 1 30.3 3 3600 5 3 24.7 5 2100 3 3 27.7 5 2900

2 1 27.4 6 2700 3 3 22.9 4 1600 3 1 25.7 5 2000 3 3 28.3 15 3000

3 3 27.2 3 2700 4 3 26.2 3 2300 3 1 27.8 0 2750 5 3 25.5 0 2250

4 3 27.1 0 2550 4 3 24.5 5 2050 4 1 27.0 3 2450 3 3 26.0 5 2150

3 3 28.0 1 2800 3 3 30.0 8 3050 3 3 29.0 10 3200 3 3 26.2 0 2400

3 1 26.5 0 1300 3 3 26.2 3 2400 4 3 25.6 7 2800 4 3 23.0 1 1650

4 3 23.0 0 1800 3 3 25.4 6 2250 4 3 24.2 0 1900 3 2 22.9 0 1600

4 2 26.0 3 2200 3 3 25.4 4 2250 4 3 25.7 0 1200 3 3 25.1 5 2100

4 2 24.5 0 2250 5 3 27.5 0 2900 4 3 23.1 0 1650 4 1 25.9 4 2550

3 3 25.8 0 2300 5 3 27.0 3 2250 3 3 28.5 0 3050 5 1 25.5 0 2750

5 3 23.5 0 1900 3 2 24.0 0 1700 3 1 29.7 5 3850 3 1 26.8 0 2550

5 3 26.7 0 2450 3 1 28.7 0 3200 4 3 23.1 0 1550 3 1 29.0 1 2800

4 3 25.5 0 2250 4 3 26.5 1 1967 4 3 24.5 1 2200 4 3 28.5 1 3000

3 3 28.2 1 2867 3 3 24.5 1 1600 3 3 27.5 1 2550 3 2 24.7 4 2550

Crab data, continued

3 1 25.2 1 2000 4 3 27.3 1 2900 3 3 26.3 1 2400 3 3 29.0 1 3100

3 3 25.3 2 1900 3 3 26.5 4 2300 3 3 27.8 3 3250 3 3 27.0 6 2500

4 3 25.7 0 2100 3 3 25.0 2 2100 3 3 31.9 2 3325 5 3 23.7 0 1800

5 3 29.3 12 3225 4 3 22.0 0 1400 3 3 25.0 5 2400 4 3 27.0 6 2500

4 3 23.8 6 1800 2 1 30.2 2 3275 4 3 26.2 0 2225 3 3 24.2 2 1650

3 3 27.4 3 2900 3 2 25.4 0 2300 4 3 28.4 3 3200 5 3 22.5 4 1475

3 3 26.2 2 2025 3 1 24.9 6 2300 2 2 24.5 6 1950 3 3 25.1 0 1800

3 1 28.0 4 2900 5 3 25.8 10 2250 3 3 27.9 7 3050 3 3 24.9 0 2200

3 1 28.4 5 3100 4 3 27.2 5 2400 3 2 25.0 6 2250 3 3 27.5 6 2625

3 1 33.5 7 5200 3 3 30.5 3 3325 4 3 29.0 3 2925 3 1 24.3 0 2000

3 3 25.8 0 2400 5 3 25.0 8 2100 3 1 31.7 4 3725 3 3 29.5 4 3025

4 3 24.0 10 1900 3 3 30.0 9 3000 3 3 27.6 4 2850 3 3 26.2 0 2300

3 1 23.1 0 2000 3 1 22.9 0 1600 5 3 24.5 0 1900 3 3 24.7 4 1950

3 3 28.3 0 3200 3 3 23.9 2 1850 4 3 23.8 0 1800 4 2 29.8 4 3500

3 3 26.5 4 2350 3 3 26.0 3 2275 3 3 28.2 8 3050 5 3 25.7 0 2150

3 3 26.5 7 2750 3 3 25.8 0 2200 4 3 24.1 0 1800 4 3 26.2 2 2175

4 3 26.1 3 2750 4 3 29.0 4 3275 2 1 28.0 0 2625 5 3 27.0 0 2625

3 2 24.5 0 2000

;

proc logistic data=crabs; model y(event=’1’)=width / link=logit aggregate scale=none lackfit;

event=1 tells SAS to model πi = P(Yi = 1) rather than
πi = P(Yi = 0). The default link is logit (giving logistic
regression) – I specify it here anyway for transparency.

aggregate scale=none lackfit give goodness-of-fit statistics,
coming up.

proc logistic output

The LOGISTIC Procedure

Model Information

Response Variable y

Number of Response Levels 2

Model binary logit

Optimization Technique Fisher’s scoring

Response Profile

Ordered Total

Value y Frequency

1 0 62

2 1 111

Probability modeled is y=1.

Deviance and Pearson Goodness-of-Fit Statistics

Criterion Value DF Value/DF Pr > ChiSq

Deviance 69.7260 64 1.0895 0.2911

Pearson 55.1779 64 0.8622 0.7761

Number of unique profiles: 66

Model Fit Statistics

Intercept

Intercept and

Criterion Only Covariates

AIC 227.759 198.453

SC 230.912 204.759

-2 Log L 225.759 194.453

SAS output, continued

Testing Global Null Hypothesis: BETA=0

Test Chi-Square DF Pr > ChiSq

Likelihood Ratio 31.3059 1 <.0001

Score 27.8752 1 <.0001

Wald 23.8872 1 <.0001

Analysis of Maximum Likelihood Estimates

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 -12.3508 2.6287 22.0749 <.0001

width 1 0.4972 0.1017 23.8872 <.0001

Odds Ratio Estimates

Point 95% Wald

Effect Estimate Confidence Limits

width 1.644 1.347 2.007

Hosmer and Lemeshow Goodness-of-Fit Test

Chi-Square DF Pr > ChiSq

5.2465 8 0.7309

14.3 Model interpretation

For simple logistic regression

Yi ∼ Bern

(
eβ0+β1xi

1 + eβ0+β1xi

)
.

An odds ratio: let’s look at how the odds of success changes when
we increase x by one unit:

π(x + 1)/[1− π(x + 1)]

π(x)/[1− π(x)]
=

[
eβ0+β1x+β1

1+eβ0+β1x+β1

]
/
[

1
1+eβ0+β1x+β1

]
[

eβ0+β1x

1+eβ0+β1x

]
/
[

1
1+eβ0+β1x

]
=

eβ0+β1x+β1

eβ0+β1x
= eβ1 .

When we increase x by one unit, the odds of an event occurring
increases by a factor of eβ1 , regardless of the value of x .

Horseshoe crab data

Let’s look at Yi = 1 if a female crab has one or more satellites,
and Yi = 0 if not. So

π(x) =
eβ0+β1x

1 + eβ0+β1x
,

is the probability of a female having more than her nest-mate
around as a function of her width x .

From SAS’s output we obtain a table with estimates β̂0 and β̂1 as
well as standard errors, χ2 test stattistics, and p-values that
H0 : β0 = 0 and H0 : β1 = 0. We also obtain an estimate of the
odds ratio eb1 and a 95% CI for eβ1 .

20 / 59



SAS output

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 -12.3508 2.6287 22.0749 <.0001

width 1 0.4972 0.1017 23.8872 <.0001

Odds Ratio Estimates

Point 95% Wald

Effect Estimate Confidence Limits

width 1.644 1.347 2.007

We estimate the probability of a satellite as

π̂(x) =
e−12.35+0.50x

1 + e−12.35+0.50x
.

The odds of having a satellite increases by a factor between 1.3
and 2.0 times for every cm increase in carapace width.

The coefficient table houses estimates β̂j , se(β̂j), and the Wald

statistic z2
j = {β̂j/se(β̂j)}2 and p-value for testing H0 : βj = 0.

What do we conclude here?

14.4 Multiple predictors

Now we have k = p − 1 predictors xi = (1, xi1, . . . , xi ,p−1) and fit

Yi ∼ bin

(
ni ,

exp(β0 + β1xi1 + · · ·+ βp−1xi ,p−1)

1 + exp(β0 + β1xi1 + · · ·+ βp−1xi ,p−1)

)
.

Many of these predictors may be sets of dummy variables
associated with categorical predictors.

eβj is now termed the adjusted odds ratio. This is how the
odds of the event occurring changes when xj increases by one
unit keeping the remaining predictors constant.

This interpretation may not make sense if two predictors are
highly related. Examples?

Categorical predictors

Have predictor X , a categorical variable that takes on values
x ∈ {1, 2, . . . , I}. Need to allow each level of X = x to affect π(x)
differently. This is accomplished by the use of dummy variables.
The usual zero/one dummies z1, z2, . . . , zI−1 for X are defined:

zj =

{
1 X = j
0 X 6= j

}
.

This is default in PROC GENMOD with a CLASS X statement; can
be obtained in PROC LOGISTIC with the PARAM=REF option.

Example with I = 3

Say I = 3, then a simple logistic regression in X is

logit π(x) = β0 + β1z1 + β2z2.

which gives

logit π(x) = β0 + β1 when X = 1

logit π(x) = β0 + β2 when X = 2

logit π(x) = β0 when X = 3

This sets class X = I as baseline. The first category can be set to
baseline with REF=FIRST next to the variable name in the CLASS
statement.

14.5 Tests of regression effects

An overall test of H0 : logit π(x) = β0 versus H1 : logit π(x) = x′β
is generated in PROC LOGISTIC three different ways: LRT, score,
and Wald versions. This checks whether some subset of variables
in the model is important.

Recall the crab data covariates:

C = color (1,2,3,4=light medium, medium, dark medium,
dark).

S = spine condition (1,2,3=both good, one worn or broken,
both worn or broken).

W = carapace width (cm).

Wt = weight (kg).

We’ll take C = 4 and S = 3 as baseline categories.

Crab data

There are two categorical predictors, C and S , and two continuous
predictors W and Wt. Let Y = 1 if a randomly drawn crab has
one or more satellites and x = (C ,S ,W ,Wt) be her covariates.
An additive model including all four covariates would look like

logit π(x) = β0 + β1I{C = 1}+ β2I{C = 2}+ β3I{C = 3}
+β4I{S = 1}+ β5I{S = 2}+ β6W + β7Wt

This model is fit via

proc logistic data=crabs1 descending;

class color spine / param=ref;

model y = color spine width weight / lackfit;

The H-L GOF statistic yields p = 0.88 so there’s no evidence of
gross lack of fit.

Parameter estimates & overall tests

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 -9.2734 3.8378 5.8386 0.0157

color 1 1 1.6087 0.9355 2.9567 0.0855

color 2 1 1.5058 0.5667 7.0607 0.0079

color 3 1 1.1198 0.5933 3.5624 0.0591

spine 1 1 -0.4003 0.5027 0.6340 0.4259

spine 2 1 -0.4963 0.6292 0.6222 0.4302

width 1 0.2631 0.1953 1.8152 0.1779

weight 1 0.8258 0.7038 1.3765 0.2407

Color seems to be important. Plugging in β̂ for β,

logit π̂(x) = −9.27 + 1.61I{C = 1}+ 1.51I{C = 2}+ 1.11I{C = 3}
−0.40I{S = 1} − 0.50I{S = 2}+ 0.26W + 0.83Wt

Overall checks that one or more predictors are important:

Testing Global Null Hypothesis: BETA=0

Test Chi-Square DF Pr > ChiSq

Likelihood Ratio 40.5565 7 <.0001

Score 36.3068 7 <.0001

Wald 29.4763 7 0.0001

Type III tests for dropping effects

Type III tests are (1) H0 : β1 = β2 = β3 = 0, color not needed to
explain whether a female has satellite(s), (2) H0 : β4 = β5 = 0,
spine not needed, (3) H0 : β6 = 0, width not needed, and (4)
H0 : β7 = 0, weight not needed:

Type 3 Analysis of Effects

Wald

Effect DF Chi-Square Pr > ChiSq

color 3 7.1610 0.0669

spine 2 1.0105 0.6034

width 1 1.8152 0.1779

weight 1 1.3765 0.2407

Largest p-value is 0.6 for dropping spine; when refitting model
without spine, still strongly reject
H0 : β1 = β2 = β3 = β4 = β5 = β6 = 0, and the H-L shows no
evidence of lack of fit. We have:

Type 3 Analysis of Effects

Wald

Effect DF Chi-Square Pr > ChiSq

color 3 6.3143 0.0973

width 1 2.3355 0.1265

weight 1 1.2263 0.2681

Continuing...

We do not reject that we can drop weight from the model, and so
we do:

Testing Global Null Hypothesis: BETA=0

Test Chi-Square DF Pr > ChiSq

Likelihood Ratio 38.3015 4 <.0001

Score 34.3384 4 <.0001

Wald 27.6788 4 <.0001

Type 3 Analysis of Effects

Wald

Effect DF Chi-Square Pr > ChiSq

color 3 6.6246 0.0849

width 1 19.6573 <.0001

Analysis of Maximum Likelihood Estimates

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 -12.7151 2.7618 21.1965 <.0001

color 1 1 1.3299 0.8525 2.4335 0.1188

color 2 1 1.4023 0.5484 6.5380 0.0106

color 3 1 1.1061 0.5921 3.4901 0.0617

width 1 0.4680 0.1055 19.6573 <.0001

29 / 59



Final model

The new model is

logit π(x) = β0 + β1I{C = 1}+ β2I{C = 2}β3I{C = 3}+ β4W .

We do not reject that color can be dropped from the model
H0 : β1 = β2 = β3, but we do reject that the dummy for C = 2
can be dropped, H0 : β2 = 0.

Maybe unnecessary levels in color are clouding its importance. It’s
possible to test whether we can combine levels of C using
contrast statements. When I tried this, I was able to combine
colors 1, 2, and 3 into one “light” category vs. color 4 “dark.”

Comments

The odds of having satellite(s) significantly increases by
e1.4023 ≈ 4 for medium vs. dark crabs.

The odds of having satellite(s) significantly increases by a
factor of e0.4680 ≈ 1.6 for every cm increase in carapace width
when fixing color.

Lighter, wider crabs tend to have satellite(s) more often.

The H-L GOF test shows no gross LOF.

We didn’t check for interactions. If an interaction between
color and width existed, then the odds ratio of satellite(s) for
different colored crabs would change with how wide she is.

Interactions

An additive model is easily interpreted because an odds ratio from
changing values of one predictor does not change with levels of
another predictor. However, often this incorrect and we may
introduce additional terms into the model such as interactions.

An interaction between two predictors allows the odds ratio for
increasing one predictor to change with levels of another. For
example, in the last model fit the odds of having satellite(s)
increases by a factor of 4 for medium crabs vs. dark regardless of
carapace width.

Interactions, continued...

A two-way interaction is defined by multiplying the variables
together; if one or both variables are categorical then all possible
pairings of dummy variables are considered.

In PROC GENMOD and PROC LOGISTIC, categorical variables
are defined through the CLASS statement and all dummy variables
are created and handled internally. The Type III table provides a
test that the interaction can be dropped; the table of regression
coefficients tell you whether individual dummies can be dropped.

Quadratic effects (pp. 575–577)

For a categorical predictor X with I levels, adding I − 1 dummy
variables allows for a different event probability at each level of X .

For a continuous predictor Z , the model assumes that the log-odds
of the event increases linearly with Z . This may or may not be a
reasonable assumption, but can be checked by adding nonlinear
terms, the simplest being Z 2.

Consider a simple model with continuous Z :

logit π(Z ) = β0 + β1Z .

LOF from this model can manifest itself in rejecting a GOF test
(Pearson, deviance, or H-L) or a residual plot that shows curvature.

Quadratic and higher order effects

Adding a quadratic term

logit π(Z ) = β0 + β1Z + β2Z 2,

may improve fit and allows testing the adequacy of the simpler
model via H0 : β2 = 0. Cubic and higher order powers can be
added, but the model can become unstable with, say, higher than
cubic powers. A better approach might be to fit a generalized
additive model (GAM):

logit π(Z ) = f (Z ),

where f (·) is estimated from the data, often using splines.

Adding a simple quadratic term can be done, e.g.,
proc logistic; model y/n = z z*z;

14.6 Model selection

Two competing goals:

Model should fit the data well.

Model should be simple to interpret (smooth rather than
overfit – principle of parsimony).

Often hypotheses on how the outcome is related to specific
predictors will help guide the model building process.

Agresti (2013) suggests a rule of thumb: at least 10 events and 10
non-events should occur for each predictor in the model (including
dummies). So if

∑N
i=1 yi = 40 and

∑N
i=1 ni = 830, you should

have no more than 40/10 = 4 predictors in the model.

Horseshoe crab data

Recall that in all models fit we strongly rejected
H0 : logit π(x) = β0 in favor of H1 : logit π(x) = x′β:

Testing Global Null Hypothesis: BETA=0

Test Chi-Square DF Pr > ChiSq

Likelihood Ratio 40.5565 7 <.0001

Score 36.3068 7 <.0001

Wald 29.4763 7 0.0001

However, it was not until we carved superfluous predictors from the
model that we showed significance for the included model effects.

This is an indication that several covariates may be highly related,
or correlated. Often variables are highly correlated and therefore
one or more are redundant. We need to get rid of some!

Automated model selection

Although not ideal, automated model selection is necessary with
large numbers of predictors. With p − 1 = 10 predictors, there are
210 = 1024 possible models; with p − 1 = 20 there are 1, 048, 576
to consider.

Backwards elimination starts with a large pool of potential
predictors and step-by-step eliminates those with (Wald) p-values
larger than a cutoff (the default is 0.05 in SAS PROC LOGISTIC).

We performed backwards elimination by hand for the crab mating
data.

Backwards elimination in SAS, default cutoff

proc logistic data=crabs1 descending;

class color spine / param=ref;

model y = color spine width weight color*spine color*width color*weight

spine*width spine*weight width*weight / selection=backward;

When starting from all main effects and two-way interactions, the
default p-value cutoff 0.05 yields only the model with width as a
predictor

Summary of Backward Elimination

Effect Number Wald

Step Removed DF In Chi-Square Pr > ChiSq

1 color*spine 6 9 0.0837 1.0000

2 width*color 3 8 0.8594 0.8352

3 width*spine 2 7 1.4906 0.4746

4 weight*spine 2 6 3.7334 0.1546

5 spine 2 5 2.0716 0.3549

6 width*weight 1 4 2.2391 0.1346

7 weight*color 3 3 5.3070 0.1507

8 weight 1 2 1.2263 0.2681

9 color 3 1 6.6246 0.0849

Analysis of Maximum Likelihood Estimates

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 -12.3508 2.6287 22.0749 <.0001

width 1 0.4972 0.1017 23.8872 <.0001

Backwards elimination in SAS, cutoff 0.15

Let’s change the criteria for removing a predictor to
p-value ≥ 0.15.
model y = color spine width weight color*spine color*width color*weight

spine*width spine*weight width*weight / selection=backward slstay=0.15;

Yielding a more complicated model:
Summary of Backward Elimination

Effect Number Wald

Step Removed DF In Chi-Square Pr > ChiSq

1 color*spine 6 9 0.0837 1.0000

2 width*color 3 8 0.8594 0.8352

3 width*spine 2 7 1.4906 0.4746

4 weight*spine 2 6 3.7334 0.1546

5 spine 2 5 2.0716 0.3549

Analysis of Maximum Likelihood Estimates

Standard Wald

Parameter DF Estimate Error Chi-Square Pr > ChiSq

Intercept 1 13.8781 14.2883 0.9434 0.3314

color 1 1 1.3633 5.9645 0.0522 0.8192

color 2 1 -0.6736 2.6036 0.0669 0.7958

color 3 1 -7.4329 3.4968 4.5184 0.0335

width 1 -0.4942 0.5546 0.7941 0.3729

weight 1 -10.1908 6.4828 2.4711 0.1160

weight*color 1 1 0.1633 2.3813 0.0047 0.9453

weight*color 2 1 0.9425 1.1573 0.6632 0.4154

weight*color 3 1 3.9283 1.6151 5.9155 0.0150

width*weight 1 0.3597 0.2404 2.2391 0.1346

AIC & model selection

“No model is correct, but some are more useful than others.” –
George Box.

It is often of interest to examine several competing models. In
light of underlying biology or science, one or more models may
have relevant interpretations within the context of why data were
collected in the first place.

In the absence of scientific input, a widely-used model selection
tool is the Akaike information criterion (AIC),

AIC = −2[L(β̂; y)− p].

The L(β̂; y) represents model fit. If you add a parameter to a
model, L(β̂; y) has to increase. If we only used L(β̂; y) as a
criterion, we’d keep adding predictors until we ran out. The p
penalizes for the number of the predictors in the model.

Crab data

The AIC has very nice properties in large samples in terms of
prediction. The smaller the AIC is, the better the model fit
(asymptotically).

Model AIC

W 198.8
C + W 197.5

C + W + Wt + W ∗ C + W ∗Wt 196.8

The best model is the most complicated one, according to AIC.
One might choose the slightly “worse” model C + W for its
enhanced interpretability.

14.7 Goodness of fit and grouping

The deviance GOF statistic is defined to be

G 2 = 2
c∑

j=1

{
Y·j log

(
Y·j

nj π̂j

)
+ (nj − Y·j) log

(
1− Y·j/nj

1− π̂j

)}
,

where π̂j = e
x′j b

1+e
x′
j
b

are fitted values.

Pearson’s GOF statistic is

X 2 =
c∑

j=1

(Y·j − nj π̂j)
2

nj π̂j(1− π̂j)
.

Both statistics are approximately χ2
c−p in large samples assuming

that the number of trials n =
∑c

j=1 nj increases in such a way that
each nj increases. These are the same type of GOF test requiring
replicates in Sections 3.7 & 6.8.

43 / 59



Aggregating over distinct covariates

Binomial data is often recorded as individual (Bernoulli) records:

i yi ni xi
1 0 1 9
2 0 1 14
3 1 1 14
4 0 1 17
5 1 1 17
6 1 1 17
7 1 1 20

Grouping the data yields an identical model:

i yi ni xi
1 0 1 9
2 1 2 14
3 2 3 17
4 1 1 20

β̂, se(β̂j), and L(β̂) don’t care if data are grouped.

The quality of residuals and GOF statistics depend on how
data are grouped. D and Pearson’s X 2 will change! (Bottom,
p. 590).

Comments on grouping

In PROC LOGISTIC type AGGREGATE and SCALE=NONE
after the MODEL statement to get D and X 2 based on
grouped data. This option does not compute residuals based
on the grouped data. You can aggregate over all variables or a
subset, e.g. AGGREGATE=(width).

The Hosmer and Lemeshow test statistic orders observations
(xi ,Yi ) by fitted probabilities π̂(xi ) from smallest to largest
and divides them into (typically) g = 10 groups of roughly the
same size. A Pearson test statistic is computed from these g
groups; this statistic is approximately χ2

g−2. Termed a
“near-replicate GOF test.” The LACKFIT option in PROC
LOGISTIC gives this statistic.

Comments on grouping

Pearson, Deviance, and Hosmer & Lemeshow all provide a
p-value for the null H0 : the model fits based on χ2

c−p where c
is the distinct number of covariate levels (using
AGGREGATE). The alternative model for the first two is the
saturated model where every µi is simply replaced by yi .

Can also test logit{π(x)} = β0 + β1x versus more general
model logit{π(x)} = β0 + β1x + β2x2 via H0 : β2 = 0.

Crab data GOF tests, only width as predictor

Raw (Bernoulli) data with aggregate scale=none lackfit;

Deviance and Pearson Goodness-of-Fit Statistics

Criterion Value DF Value/DF Pr > ChiSq

Deviance 69.7260 64 1.0895 0.2911

Pearson 55.1779 64 0.8622 0.7761

Number of unique profiles: 66

Partition for the Hosmer and Lemeshow Test

y = 1 y = 0

Group Total Observed Expected Observed Expected

1 19 5 5.39 14 13.61

2 18 8 7.62 10 10.38

3 17 11 8.62 6 8.38

4 17 8 9.92 9 7.08

5 16 11 10.10 5 5.90

6 18 11 12.30 7 5.70

7 16 12 12.06 4 3.94

8 16 12 12.90 4 3.10

9 16 13 13.69 3 2.31

10 20 20 18.41 0 1.59

Hosmer and Lemeshow Goodness-of-Fit Test

Chi-Square DF Pr > ChiSq

5.2465 8 0.7309

Comments

There are c = 66 distinct widths {xi} out of n = 173 crabs.
For χ2

66−2 to hold, we must keep sampling crabs that only
have one of the 66 fixed number of widths! Does that make
sense here?

The Hosmer and Lemeshow test gives a p-value of 0.73 based
on g = 10 groups.

The raw statistics do not tell you where lack of fit occurs.
Deviance and Pearson residuals do tell you this (later). Also,
the table provided by the H-L tells you which groups are ill-fit
should you reject H0 : logistic model holds.

GOF tests are meant to detect gross deviations from model
assumptions. No model ever truly fits data except
hypothetically.

14.8 Diagnostics

GOF tests are global checks for model adequacy. Residuals and
influential measures can refine a model inadequacy diagnosis.

The data are (xj ,Y·j) for j = 1, . . . , c . The j th fitted value is an

estimate of µj = E (Y·j), namely Ê (Y·j) = µ̂j = nj π̂j where

πj = e
β′xj

1+e
β′xj

and π̂j = e
β̂
′
xj

1+e
β̂
′
xj

. The raw residual ej is what we see

Y·j minus what we predict nj π̂j . The Pearson residual divides this
by an estimate of

√
var(Y·j):

rPj
=

y·j − nj π̂j√
nj π̂j(1− π̂j)

.

The Pearson GOF statistic is

X 2 =
c∑

j=1

r2
Pj
.

Diagnostics

The standardized Pearson residual is given by

rSPj
=

y·j − nj π̂j√
nj π̂j(1− π̂j)(1− ĥj)

,

where ĥj is the j th diagonal element of the hat matrix

Ĥ = Ŵ1/2X(X′ŴX)−1X′Ŵ1/2 where X is the design matrix

X =


1 x11 · · · x1,p−1

1 x21 · · · x2,p−1

...
...

. . .
...

1 xc1 · · · xc,p−1

 ,

and

Ŵ =


n1π̂1(1− π̂1) 0 · · · 0

0 n2π̂2(1− π̂2) · · · 0
...

...
. . .

...
0 0 · · · nc π̂c (1− π̂c )

 .
Alternatively, p. 592 defines a deviance residual.

Diagnostics

Your book suggests lowess smooths of residual plots (pp.
594–595), based on the identity E (Yi − π̂i ) = E (ei ) = 0 for
Bernoulli data. You’ll consider these for your homework; you
are looking for a line that is approximately zero, not perfectly
zero. The line will have a natural increase/decrease at either
end if there are lots of zeros or ones – e.g. last two plots on p.
595.
Residual plots for individual predictors might show curvature;
adding quadratic terms or interactions can improve fit.
An overall plot is a smoothed rSPj

versus the linear predictor

η̂j = β̂
′
xj . This plot will tell you if the model tends to over or

underpredict the observed data for ranges of the linear
predictor.
You can look at individual rSPj

to determine model fit. For
the crab data, this might flag some individual crabs as ill-fit or
unusual relative to the model. I usually flag |rSPj

| > 3 as
being ill-fit by the model.

Influential observations

Unlike linear regression, the leverage ĥj in logistic regression

depends on the model fit β̂ as well as the covariates xj . Points
that have extreme predictor values xj may not have high leverage

ĥj if π̂j is close to 0 or 1. Here are the influence diagnostics
available in PROC LOGISTIC:

Leverage ĥj . Still may be useful for detecting “extreme”
predictor values xj .

cj = r2
SPj

ĥj/[p(1− ĥj)
2] measures the change in the joint

confidence region for β when j is left out (Cook’s distance).

DFBETAjs is the standardized change in β̂s when observation
j is left out.

The change in the X 2 GOF statistic when obs. j is left out is
DIFCHISQj = r2

SPj
/(1− ĥj). (∆X 2

j in your book)

I suggest simply looking at plots of cj vs. j .

Diagnostics/influence in crab data

proc logistic data=crabs descending; class color / param=ref;

model y(event=’1’)=color width;

output out=diag1 stdreschi=r xbeta=eta p=p c=c;

proc sgscatter data=diag1;

title "Crab data diagnostic plots";

plot r*(width eta) r*color / loess;

proc sgscatter data=diag1;

title "Crab data diagnostic plots";

plot (r c)*id;

proc sort; by color width;

proc sgplot data=diag1;

title1 "Predicted probabilities";

series x=width y=p / group=color;

yaxis min=0 max=1;

proc print data=diag1(where=(c>0.3 or r>3 or r<-3));

var y width color c r;

Case study, mosquito infection (pp. 573–575)

Study of a disease outbreak spread by mosquitoes. Yi is whether
the ith person got the disease. xi1 is the person’s age, xi2 is
socioeconomic status (1=upper, 2=middle, 3=lower), and xi3 is
sector (0=sector 1 and 1=sector 2).

filename mosq url ’http://www.stat.sc.edu/~hansont/stat705/mosquito_full.txt’;

data disease;

infile mosq;

input case age ses sector disease dummy;

run;

proc logistic data=disease;

class ses sector / param=ref;

model disease(event=’1’)=age ses sector / lackfit;

Note smoothed residual plot vs. age! Try backwards elimination
from full interaction model.

Fitting logistic regression models (pp. 564–565)

The data are (xj ,Y·j) for j = 1, . . . , c .
The model is

Y·j ∼ bin

(
nj ,

eβ′xj

1 + eβ′xj

)
.

The pmf of Y·j in terms of β is

p(yj ; β) =

(
nj

y·j

)[
eβ′xj

1 + eβ′xj

]y·j [
1− eβ′xj

1 + eβ′xj

]nj−y·j

.

The likelihood is the product of all N of these and the
log-likelihood simplifies to

L(β) =

p∑
k=1

βk

c∑
j=1

y·jxjk−
c∑

j=1

log

[
1 + exp

(
p∑

k=1

βkxjk

)]
+constant.

Inference

The likelihood (or score) equations are obtained by taking partial
derivatives of L(β) with respect to elements of β and setting equal
to zero. Newton-Raphson is used to get β̂, see the following
optional slides if interested.

The inverse of the covariance of β̂ has ij th element

−∂
2L(β)

∂βi∂βj
=

N∑
s=1

xsixsjnsπs(1− πs),

where πs = eβ′xs

1+eβ′xs . The estimated covariance matrix ĉov(β̂) is

obtained by replacing β with β̂. This can be rewritten

ĉov(β̂) = {X′diag[nj π̂j(1− π̂j)]X}−1.

How to get the estimates? (Optional...)

Newton-Raphson in one dimension: Say we want to find where
f (x) = 0 for differentiable f (x). Let x0 be such that f (x0) = 0.
Taylor’s theorem tells us

f (x0) ≈ f (x) + f ′(x)(x0 − x).

Plugging in f (x0) = 0 and solving for x0 we get x̂0 = x − f (x)
f ′(x) .

Starting at an x near x0, x̂0 should be closer to x0 than x was.
Let’s iterate this idea t times:

x (t+1) = x (t) − f (x (t))

f ′(x (t))
.

Eventually, if things go right, x (t) should be close to x0.

Higher dimensions

If f(x) : Rp → Rp, the idea works the same, but in vector/matrix
terms. Start with an initial guess x(0) and iterate

x(t+1) = x(t) − [Df(x(t))]−1f(x(t)).

If things are “done right,” then this should converge to x0 such
that f(x0) = 0.
We are interested in solving DL(β) = 0 (the score, or likelihood
equations!) where

DL(β) =


∂L(β)
∂β1
...

∂L(β)
∂βp

 and D2L(β) =


∂L(β)
∂β2

1
· · · ∂L(β)

∂β1∂βp

...
. . .

...
∂L(β)
∂βp∂β1

· · · ∂L(β)
∂β2

p

 .

Newton-Raphson

So for us, we start with β(0) (maybe through a MOM or least
squares estimate) and iterate

β(t+1) = β(t) − [D2L(β)(β(t))]−1DL(β(t)).

The process is typically stopped when |β(t+1) − β(t)| < ε.

Newton-Raphson uses D2L(β) as is, with the y plugged in.

Fisher scoring instead uses E{D2L(β)}, with expectation
taken over Y, which is not a function of the observed y, but
harder to get.

The latter approach is harder to implement, but conveniently
yields ĉov(β̂) ≈ [−E{D2L(β)}]−1 evaluated at β̂ when the
process is done.

